

Die Funktion ist wahrlich etwas knifflig und ich bin auch nicht allein darauf gekommen, sondern habe mir in verschiedenen Forenbeiträgen Anregungen geholt:

=HYPERLINK(TEIL(ZELLE("filename");2;SUCHEN("/[^\"]*";ZELLE("filename"))-1) & A1 & ".pdf";A1& ".pdf")

Sie setzt sich schrittweise aus mehreren Funktionen zusammen:

ZELLE("filename")

liefert:

[file:///home/michael/.opera/temporary_downloads/hyperlink.ods#\\$Tabelle1](file:///home/michael/.opera/temporary_downloads/hyperlink.ods#$Tabelle1)

die URL des aktuellen Dokuments

SUCHEN("/[^\"]*";ZELLE("filename"))

liefert:

49

die Position von / gefolgt von beliebigen Zeichen außer / und einem ' (m. a. W. die Position im Text wo der eigentliche Dateiname beginnt und der Pfad dahin endet)

hier werden reguläre Ausdrücke zur Suche verwendet - Genaueres siehe dort

TEIL(ZELLE("filename");2;SUCHEN("/[^\"]*";ZELLE("filename"))-1)

liefert

file:///home/michael/.opera/temporary_downloads/

aus der ganzen URL wird der Teil ab dem 2. Zeichen und zwar 48 Zeichen ausgeschnitten

TEIL(ZELLE("filename");2;SUCHEN("/[^\"]*";ZELLE("filename"))-1) & A1 & ".pdf"

liefert

file:///home/michael/.opera/temporary_downloads/latschen.pdf

der Pfad wird mit dem Text aus A1 („latschen“) und „.pdf“ verkettet

HYPERLINK(TEIL(ZELLE("filename");2;SUCHEN("/[^\"]*";ZELLE("filename"))-1) & A1 & ".pdf";A1& ".pdf")

erzeugt einen Hyperlink auf die Datei und zeigt in der Zelle B1 den Text „latschen.pdf“ an